


```
type
  t_indice = 0..TAILLE_MAX ;
  t_bit = boolean ;
  t_tabImage = array[t_indice, t_indice] of t_bit ;
  t_image = record
    tab : t_tabImage ;
    largeur : t_indice ;
    hauteur : t_indice ;
  end ;
```

Question 2

Écrire un sous programme **lireImage** qui permette de transférer le contenu d'un fichier image dans une variable de type **t_image**.

```
procedure lireImage(nomFichier : string ; var img : t_image);
var
  f : text ;
  c : char ;
  i, j : t_indice ;
begin
  assign(f, nomFichier) ;
  reset(f);
  i := 0;
  while (not eof(f)) do
    begin
      i := i+1;
      j := 0;
      while (not eoln(f)) do
        begin
          j := j+1;
          read(f,c);
          img.tab[i,j] := c = '1';
        end ;
      read(f,c);
    end ;
  img.largeur := j;
  img.hauteur := i ;
  close(f);
end ;
```

Question 3

Écrire un sous-programme **afficheImage** affiche pour afficher une telle image à l'écran. Les '1' correspondront à l'écriture d'un 'X' et les '0' seront traduits par un espace.

```
procedure afficheImage(img : t_image);
var
  i,j : t_indice;
begin
```

```

for i := 1 to img.hauteur do
begin
  for j := 1 to img.largeur do
    if img.tab[i,j] then
      write ('X')
    else
      write (' ');
  writeln ;
end ;
end ;

```

Question 4

Écrire un sous-programme **ecrireImage** qui enregistre une image de type **t_image** dans un fichier dont le nom est passé en argument.

```

procedure ecrireImage(fichier : string ; img : t_image);
var
  f : text;
  i,j : t_indice;
begin
  assign(f, fichier) ;
  rewrite(f);
  for i := 1 to img.hauteur do
  begin
    for j := 1 to img.largeur do
      if img.tab[i,j] then
        write (f,'1')
      else
        write (f,'0');
    writeln (f,"");
  end;
  close(f);
end;

```

Question 5

Écrire un sous-programme **negatif** qui transforme une image en son négatif, c'est à dire qui remplace chaque '1' par un '0' et vice versa.

```

procedure negatif(var img : t_image);
var
  i,j : t_indice;
begin
  for i := 1 to img.hauteur do
    for j := 1 to img.largeur do
      img.tab[i,j] := not img.tab[i,j] ;
  end;

```

Question 6



Écrire un sous programme **erosion** qui ne conserve à '1' que les points dont les 8 voisins étaient à '1' avant cette transformation. Attention cette transformation (comme les suivantes) nécessite l'emploi d'un tableau intermédiaire.

```

procedure erosion(var img : t_image);
var
  tmp : t_tabImage;
  i,j : t_indice;
begin
  tmp := img.tab ;
  for i := 2 to img.hauteur-1 do
    for j := 2 to img.largeur-1 do
      tmp[i,j] := img.tab[i,j] and
        img.tab[i-1,j-1] and
        img.tab[i-1,j] and
        img.tab[i-1,j+1] and
        img.tab[i,j-1] and
        img.tab[i,j+1] and
        img.tab[i+1,j-1] and
        img.tab[i+1,j] and
        img.tab[i+1,j+1] ;
  img.tab := tmp ;
end;

```

Question 7

Écrire un sous programme **dilatation** qui remplace chaque '0' par un '1' dès lors qu'au moins un de ses 8 voisins est à '1'.

```

procedure dilatation(var img : t_image);
var
  tmp : t_tabImage;
  i,j : t_indice;
begin
  tmp := img.tab ;
  for i := 2 to img.hauteur-1 do
    for j := 2 to img.largeur-1 do
      tmp[i,j] := img.tab[i,j] or
        img.tab[i-1,j-1] or
        img.tab[i-1,j] or
        img.tab[i-1,j+1] or
        img.tab[i,j-1] or
        img.tab[i,j+1] or
        img.tab[i+1,j-1] or
        img.tab[i+1,j] or
        img.tab[i+1,j+1] ;
  img.tab := tmp ;
end;

```

Question 8

Écrire un sous programme **supprimerBruit** qui supprime les points isolés, c'est-à-dire qui remplace un

'1' par un '0' lorsque ce '1' est entouré de 8 '0' et qui remplace un '0' par un '1' quand le '0' est entouré de 8 '1'.

```

procedure supprimerBruit(var img : t_image);
var
  i,j : t_indice;
begin
  for i := 2 to img.hauteur-1 do
    for j := 2 to img.largeur-1 do
      img.tab[i,j] := img.tab[i,j] xor (
        (img.tab[i-1,j-1] = not img.tab[i,j]) and
        (img.tab[i-1,j] = not img.tab[i,j]) and
        (img.tab[i-1,j+1] = not img.tab[i,j]) and
        (img.tab[i,j-1] = not img.tab[i,j]) and
        (img.tab[i,j+1] = not img.tab[i,j]) and
        (img.tab[i+1,j-1] = not img.tab[i,j]) and
        (img.tab[i+1,j] = not img.tab[i,j]) and
        (img.tab[i+1,j+1] = not img.tab[i,j])
      )
    end;

```

Question 9

Écrire un sous programme **ouverture** qui ne conserve dans l'image que les '1' appartenant à tous les groupes de 4 '1' en carré, tous les autres points sont à '0'.

```

procedure ouverture(var img : t_image);
var
  tmp : t_tabImage;
  i,j : t_indice;
begin
  for i := 1 to img.hauteur do
    for j := 1 to img.largeur do
      tmp[i,j] := false;
  for i := 1 to img.hauteur-1 do
    for j := 1 to img.largeur-1 do
      if img.tab[i,j] and img.tab[i,j+1] and img.tab[i+1,j] and img.tab[i+1,j+1] then
        begin
          tmp[i,j] := true ;
          tmp[i+1,j] := true ;
          tmp[i,j+1] := true ;
          tmp[i+1,j+1] := true ;
        end;
  img.tab := tmp ;
end;

```

Question 10

Écrire un programme qui teste les sous-programmes et les déclarations précédents. Ce programme pourra avoir l'interface suivante :

Menu :
q->quitter



```

l->lire
s->sauver
a->afficher
n->negatif
e->erosion
d->dilatation
b->bruit
o->ouverture:

```

```

var
  c : char ;
  s : string ;
  img : t_image ; begin
  writeln ('Menu : ');
  writeln (' q->quitteri');
  writeln (' l->lire');
  writeln (' s->sauver');
  writeln (' a->afficher');
  writeln (' n->negatif');
  writeln (' b->bruit');
  writeln (' e->erosion');
  writeln (' d->dilatation');
  writeln (' o->ouverture');
  repeat
    write ('Entrer une action : ');
    readln (c);
    case c of
      'l' : begin
        write ('Entrer le nom du fichier a ouvrir : ');
        readln (s);
        lireImage(s,img);
      end;
      's' : begin
        write ('Entrer un nom pour enregistrer le fichier : ');
        readln (s);
        ecrireImage(s,img);
      end;
      'a' :
        afficheImage(img);
      'n' :
        negatif(img);
      'b' :
        supprimerBruit(img);
      'e' :
        erosion(img);
      'd' :
        dilatation(img);
      'o' :
        ouverture(img);
    end;
  until (c = 'q');
end.

```